Paper 4697-2020 Model-Based Risk-Adjustment in Clinical Outcome Research

Jiming Fang, Feng Qiu ICES, Toronto, Ontario, Canada

ABSTRACT

To compare clinical outcome across patient groups or health providers, we need to account for the variations from patient-level demographic and clinical treatments. Model-based riskadjustment approach provides strategy to control these variations, which is commonly used in hospital report cards. In the present paper, we will focus on the concept of the modelbased risk-adjusted outcomes using different regression models, including multilevel hierarchical model, according to the type of outcomes.

INTRODUCTION

Health outcomes usually refer to clinically relevant endpoints of care, such as mortality or major morbidity. The outcomes may be binary in nature (e.g., death within 30-days of index date), or time-to-event (e.g., time to death), or continuous ones (e.g., length of stay at hospital). Outcome measurement is important to determine the quality of care rendered.

In **practice, we can't simply compare the crude outcome (e.g.,** 30-day mortality) across patient groups (e.g., income quintiles, races) or health providers (e.g., hospitals, geographic regions, countries), because patient-level demographic variation may exist across different population or within the same population over time. In addition, variation in clinical treatments, health care providers, plus random chance or "noise" also occurs in the health care system (Figure 1).



Figure 1. The impact of health outcomes

To account for these variations, one approach is to study the stratum-specific rates (e.g., to compare outcomes across sub-groups with same sex and same age). This approach allows us to understand the variation across strata (e.g., comparing outcomes in young men across hospitals). However, a single summary index is preferred for overall comparison. The term "standardization" refers to procedures for facilitating the comparison of summary measures across groups that are based on comparing outcomes in strata of similar subjects.

A common strategy to control confounding is stratification. Start with subdividing the population into several strata that are defined by levels of the confounding variables such as sex and age groups. Estimate the outcome of interest within each stratum. Combine the stratum-specific results into an overall estimate. However standardized rates constructed using standardization does not completely adjust for differences in population composition, but only for a few factors such as sex and age groups. Model-based risk-adjustment provides another strategy to control multiple confounders, which is commonly used in hospital report cards. In the present paper, we will focus on the concept of the model-based risk-adjusted outcomes using different regression models, according to the type of outcomes.

MODEL-BASED RISK-ADJUSTMENT

Model-based risk-adjustment is a method used to account for the impact of individual risk factors — such as age, sex, severity of illnesses, and other medical problems. In a multivariable regression model, patient risk factors can be added to control for their contribution to the outcome of interest. Thus, the providers that care for more severe **patients won't be unfairly** penalized.

When outcomes are binary, the result of the application of model-based risk-adjustment is an indirectly-standardized rate which is defined as:

$$Adjusted Rate = \frac{Observed Rate (O)}{Expected Rate (E)} \times Crude Overall Rate$$

The expected rate is the model-based estimate of the rate of the outcome conditional on the case-mix of the patients in the given cluster (hospital or region). O/E is also called risk standardized ratio. See more on concept of risk-adjustment in the book edited by lezzoni (2013, pages 196, 342-349).

The model-based risk-adjustment has received increasing attention and this method has been used in more and more in clinical studies in the past 50 years (Figure 2, **data** as of January 29, 2020 on <u>https://www.ncbi.nlm.nih.gov/pubmed/</u>).

To produce timely results for analysts, we created a user-friendly SAS macro (%mbram) to compute the <u>model-based risk-adjustment</u> results in a table and display the variation across health care providers in a visual format. It permits use of the following types of outcomes:

- Binary outcome in a specified time period (i.e., 30-day mortality, the presence of a complication)
- Binary outcome in a specified period of time with individual follow-up person-time (e.g., mortality rate per 100 person-years)
- Continuous outcome (e.g., LOS in days, cost of health care).



Figure 2. Number of PubMed papers with keywords of "risk-adjustment"

ADJUSTED BINARY OUTCOME VIA LOGISTIC MODEL

For binary outcomes, the confidence intervals of O/E ratios were computed based on the normal approximation to the binomial distribution (Hosmer 1995; Austin 2003).

$$E_i = \sum_{j=1}^{n_i} p_{ij}$$

where E_i denotes the expected outcomes for the ith provider, n_i is the number of patients treated at the ith provider, and p_{ij} denotes the probability of the outcome for the jth patient treated at the ith provider based on the logistic regression model.

$$O_i = \sum_{j=1}^{n_i} Y_{ij}$$

where O_i denotes the observed outcomes for the ith provider and Y_{ij} denotes each **patient's crude outcome (1 or 0) at the i**th provider.

$$CI_{lower i} = \frac{1}{E_i} \times max \left\{ 0, O_i - 1.96 \sqrt{\sum_{j=1}^{n_i} p_{ij} (1 - p_{ij})} \right\}$$
$$CI_{upper i} = \frac{1}{E_i} \times min \left\{ n_i, O_i + 1.96 \sqrt{\sum_{j=1}^{n_i} p_{ij} (1 - p_{ij})} \right\}$$

where $CI_{lower i}$ and $CI_{upper i}$ denotes the 95% conference interval for the ith provider's risk-adjusted O/E ratio.

A hospital was considered to have outlier performance if the 95% confidence interval around its O/E ratio excluded 1.

Here are the SAS codes to compute the expected probabilities of outcome from multivariable logistic regression model, where &adj_cvar and &adj_nvar are the macro variable for list of categorical variables and continuous variables, respectively.

```
*** Expected Risk;
Proc Genmod data=mydata;
    class &adj_cvar;
    model outcome=&adj_nvar &adj_cvar/d=b link=logit;
    output out=_exp predicted=expected_risk;
run;
```

Notice the SAS procedure above **only computes each subject's probability of the outcome,** but not the O/E ratio. Analysts need to run another SAS procedure (Proc Means or Proc Freq) to obtain the observed outcomes. And they would need to sum the observed and predicted outcomes within each cluster or group to get the O/E ratios.

ADJUSTED PERSON-TIME RATES VIA POISSON MODEL

To compute risk-adjusted rates (say, the readmission rate in 100 person-year), the analysts can compute the expected counts using multivariable Poisson regression model, with defined log of person-time. Again, analysts need to compute observed rates in order to get O/E ratio. The 95%CI of rates can be obtained using bootstrapping method.

```
*** Expected Rate;
Proc Genmod data=mydata;
    class &adj_cvar;
    model outcome=&adj_nvar &adj_cvar/d=P link=log offset=log_persontime;
    output out=_exp predicted=expected_rate;
run;
*** Observed Rate;
Proc Genmod data=mydata;
    class Hospital;
    model outcome=Hospital/d=P link=log offset=log_persontime;
    lsmean Hospital /ilink;
run;
```

ADJUSTED CONTINUOUS OUTCOME VIA GENERALIZED REGRESSION MODEL

For continuous outcome such as length of stay in days, cost in dollars, we could get the expected continuous outcome using multivariable regression model. The distribution of continuous outcomes can be normal, Poisson, negative binomial, or Gamma. The analysts need to select one of them according to model performance of these non-nested regression models, by checking Akaike's information criterion (AIC, smaller is better) or using <u>Vuong</u>

#SASGF

test approach to get p-value to identify which model fits the data significantly better. Again, analysts need to compute observed outcome in order to get O/E ratio. The 95%CI can be obtained using bootstrapping method.

```
*** Expected Outcome: Normal distribution;
Proc Genmod data=mydata descending;
      class &adj cvar;
      model outcome=&adj_nvar &adj_cvar/d=normal link=identity;
      output out=_exp predicted=expected;
run;
*** Expected Outcome: Poisson distribution;
Proc Genmod data=mydata descending;
      class &adj_cvar;
      model outcome=&adj nvar &adj cvar/d=P link=log;
      output out= exp predicted=expected;
run;
*** Expected Outcome: Negative Binomial distribution;
Proc Genmod data=mydata descending;
      class &adj_cvar;
      model outcome=&adj_nvar &adj_cvar/d=NB link=log;
      output out=_exp predicted=expected;
run;
*** Expected Outcome: Gamma distribution;
*** The log link function is selected to ensure that the mean is positive;
Proc Genmod data=mydata descending;
      class &adj_cvar;
      model outcome=&adj_nvar &adj_cvar/d=Gamma link=log;
      output out=_exp predicted=expected;
run;
```

ADJUSTED SURVIVAL RATES VIA COX MODEL

For time-to-event outcomes, at a given time of interest (e.g., 1-year from index dates), we could estimate the expected probability of the outcome occurring from a multivariable Cox proportion hazards model (Proc Phreg). Here is the SAS code to get expected time-to-event outcome probability from Cox model.

```
*** Expected survival probability;
Proc Phreg data=mydata;
    class &adj_cvar;
    Model survTime*status(0)= &adj_nvar &adj_cvar /ties=efron;
    Baseline covariates=mydata out=_Exp survival=survival timelist=365;
Run;
```

This will create a dataset called '_Exp' that contains a survival curve for each subject. Choose an event time of interest via option "timelist" in Baseline statement to output only the survival function at the given time value(s), otherwise the whole survival curve will be outputted (it will be a very long dataset). For each subject, determine the probability of surviving to that time from the dataset. Then 1 – this probability is the probability of the outcome occurring prior to this duration of follow-up. Having this probability for each patient

in the cohort allow analysts to compute the expected probability by study groups at the given time.

From the crude survival function in Proc Lifetest (i.e., KM curve), analysts can get the observed time-to-event rates at given time of interest by study groups. Based on the expected rate and observed rates, plus the overall crude rate, analysts can get the point estimation of adjusted rate at the given time. Again, bootstrapping approach is needed to estimate the corresponding 95%CI.

```
*** Observed survival probability by hospital;
Proc Lifetest data=mydata outsurv=_Obs noprint timelist=365 reduceout;
    time survTime*status(0);
    strata hospital;
```

run;

ADJUSTED SURVIVAL RATES VIA FINE-GRAY MODEL

If we need to account for competing risk factor in survival analysis, we use a Fine-Gray model to obtain the expected probability of the outcome, and use cumulative incidence function (i.e., CIF curve) to get observed probability of the outcome. Here is the SAS code to get expected and observed CIF.

```
*** Expected CIF;
Proc phreg data=mydata;
        class &adj_cvar;
        model survTime*Status(0)= &adj_nvar &adj_cvar /eventcode=1;
        baseline covariates=mydata out=_ExpCIF CIF=predCIF;
run;
*** Observed CIF by hospital;
Proc lifetest data=mydata outcif=_ObsCIF noprint;
        time survTime*Status(0) /eventcode=1;
        strata hospital;
run;
```

MODEL-BASED ADJUSTED OUTCOME VIA HIERARCHICAL MODEL

A hierarchical data structure can exist in clinical studies, for example, patients are nested within providers (e.g., hospitals, geographic regions). The nested observations may be correlated. Hierarchy model provides a comprehensive framework for accounting for variation at different levels of analysis.

Adjusted binary outcome via hierarchical Logistic model

Using the above methods, for a small hospital with zero observed events, we would have a risk-adjusted rate of zero (since the observed rate is equal to zero). Hospital with a small number of cases may have observed outcomes at the extreme ends of the range, but such outcomes **may not accurately reflect their "true" performance.**

To deal with small sample sizes and/or low event cases, we can use another method based on the use of a hierarchical logistic regression model to account for patient-specific demographic and clinical characteristics as well as the clustering of patients within the same

hospital using hospital-specific random effects (Austin 2003; Shahian 2005). Thus, the predicted number of events could differ from zero even if the observed number of events is equal to zero. In general, this approach had a lower rate of correctly classifying hospital performance that methods based on O/E ratios.

The observed outcome is replaced by predicted risk when computing risk-adjusted outcome using a hierarchical model. I.e., P/E ratio is a modification of the above O/E ratio above.

$Adjusted Risk = \frac{Predicted Risk (P)}{Expected Risk (E)} \times Crude Overall Risk$

Predicted risk (P): Using the fitted random effects logistic regression model, including the average intercept and the hospital-specific random effects, the predicted probability of the occurrence of the outcomes is estimated for each patient. Sum these predicted probabilities to get the predicted number of events at each hospital.

Predicted risk =
$$\frac{e^{\alpha_o + \alpha_{jo} + \beta x}}{1 + e^{\alpha_o + \alpha_{jo} + \beta x}}$$

where α_0 is the average intercept, α_{0} is the hospital-specific random effect and βx is the regression parameters.

Expected risk (E): The expected risk at the given hospital is based on the basis of the casemix of its patients, and assuming that the hospital had the same performance as an average hospital (i.e., dropping the hospital-specific random effect but retaining the average intercept).

Expected risk = $\frac{e^{\alpha_o + \beta x}}{1 + e^{\alpha_o + \beta x}}$

where α_0 is the average intercept and βx is the regression parameters.

Sum these predicted probabilities to get expected number of events at each hospital. The risk-adjusted outcome is then calculated, which is the ratio of predicted to expected outcome multiplied by the unadjusted overall risk.

Here are the SAS codes to compute both predicted and expected probabilities.

```
Proc glimmix data=mydata;
      class Hospital &adj_cvar;
      model outcome (event='1')= &adj_nvar &adj_cvar /d=binary;
      output out= exp
            /*blup: uses the predictors of the random effects in computing
            the statistic*/
            /*ilink: compute probability on the scale of the data*/
            pred(blup ilink)=predicted_prob
            /*noblup: does not use the predictors of the random effects in
            computing the statistic*/
            pred(noblup ilink)=expected_prob;
      random intercept /subject=Hospital;
```

run;

Adjusted person-time rates via hierarchical Poisson model

Analysts can also estimate the risk-adjusted rates via hierarchical model via Proc Glimmix by specifying Poisson distribution and log of person-time.

```
Proc glimmix data=mydata;
      class Hospital &adj cvar;
      model outcome (event='1')= &adj_nvar &adj_cvar /d=P link=log
                                 offset=log_persontime;
      output out=_exp
            pred(blup ilink) =predicted_rate
            pred(noblup ilink)=expected_rate;
      random intercept /subject=Hospital;
```

run;

Adjusted continuous outcome via hierarchical generalized regression model

We could also get the expected continuous outcomes using random effects models. The distribution can be normal, Poisson, negative binomial, gamma. Here is an example to using gamma distribution.

```
*** Outcome is LOS in days;
Proc glimmix data=mydata;
      class Hospital &adj_cvar;
      model LOS = &adj_nvar &adj_cvar /d=Gamma link=log;
      output out=_exp
            pred(blup ilink) =predicted rate
            pred(noblup ilink)=expected rate;
      random intercept /subject=Hospital;
```

run;

We could use either Bootstrapping or Bayesian analysis to assess the confidence intervals. However, it is not clear what method would be best to use for estimating CIs for the P/E ratios. Until further research is done, we will not describe the methods for computing confidence intervals in the present paper.

NOTES ON MODEL-BASED RISK-ADJUSTMENT

First, depending on the context, it may be inappropriate to adjust for certain providers' or patients' attributes. For example, if the goal is to check inequalities across hospitals, we do not adjust for hospital characteristics in the regression model and also do not adjust groupvariable related variable (e.g., hospital volume, teaching status) in the model. Otherwise, adjusting for hospitals and their characteristics would mask their differences. Take patients' income guintiles as another example, if the goal is to use risk-adjusted outcomes information to examine equalities across social economic status, adjusting for income guintiles would mask their differences.

Second, we may classify the hospital whose risk-adjusted rates are significantly different from the population average as outliers, but we could not directly compare one institution with another based their risk-adjusted outcomes.

Third, analysts need to check the model performance before computing the adjusted outcomes. The model performances include explained variation measured by R-squared,

discrimination measured by c-statistics and calibration (lezzoni 2013; Forthman 2010). If the model does not fit data well, the results of adjusted outcomes might be biased. Notice we could not evaluate the risk-adjustment model mainly based on its c-statistic because it had a very modest impact on the accuracy of the report cards of health care providers (Austin 2013; Merkow 2012).

Fourth, when both hospital volume and number were low, the P/E ratio was slightly more accurate than the O/E ratio. In general, random-effects (hierarchical) models had greater specificity than the fixed-effect logistic models, whereas the latter had greater sensitivity to detect outliers (Austin 2003). For hospital profiling, some have suggested that the use of hierarchical Logistic regression model is preferable (Shahian 2005).

Finally, be aware of the presence of unobserved/unmeasured patient characteristics. If these unknown factors have strong impact on an outcome of interest, model-based risk adjustment becomes more challenging.

REFERENCES

Austin PC, Alter DA and Tu JV. **2003.** "The use of fixed- and random-effects models for classifying hospitals as mortality outliers: a Monte Carlo assessment". *Medical Decision Making*, 23: 526-39.

Austin PC and Reeves MJ. **2013.** "The relationship between the c-statistic of a riskadjustment model and the accuracy of hospital report cards: a Monte Carlo study". *Medical Care*, 51:275-84

Forthman MT, Gold RS, Dove HG, Henderson RD. **2010.** "Risk-adjusted indices for measuring the quality of inpatient care". *Quality Management in Health Care*, 19:265-77.

Hosmer DW and Lemshow S. **1995.** "Confidence interval estimates of an index of quality performance based on Logistic regression models". *Statistics in Medicine*, 14:2161-72.

Iezzoni LI. 2013. *Risk Adjustment for Measuring Healthcare Outcomes* (4th Ed). Chicago, III.: Health Administration Press.

Merkow RP, Hall BL, Cohen ME, et al. **2012.** "Relevance of the C-Statistic When Evaluating Risk-Adjustment Models in Surgery". *Journal of the American College of Surgeons*, 214:822-30

Shahian DM, Torchiana DF, Shemin RJ et al. **2005.** "Massachusetts cardiac surgery report card: Implications of statistical methodology". *The Annals of Thoracic Surgery*, 80:2106-13

CONTACT INFORMATION

Jiming Fang, PhD ICES 2075 Bayview Avenue, G106 Toronto, Ontario M4N 3M5 Canada E-mail: jiming.fang@ices.on.ca

Feng Qiu, MSc ICES 2075 Bayview Avenue, G106 Toronto, Ontario M4N 3M5 Canada E-mail: <u>feng.qiu@ices.on.ca</u>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.